

# **New features in the closed orbit correction procedures in the MAD-X program**

**W. Herr, AB Division, CERN, 1211-Geneva 23**

## **Abstract**

The closed orbit correction procedures implemented in MAD-X have been significantly extended and allow more flexibility. In particular it is possible to correct the closed orbit to a specified target orbit rather than to the smallest possible orbit. Furthermore, any model can be used for the correction, in particular the model and the orbit to be corrected can be calculated separately.

Geneva, Switzerland

1. May 2003

# 1 Introduction

In a recent note [1] the implementation of orbit correction procedures in the MAD-X program [2] was described. The algorithms used there follow closely those available in the closed orbit correction package COCU [3, 4, 5] which was used in the control system of SPS and LEP. However the nature of MAD-X as a modelling and design program implies some inherent differences:

- The perturbed orbit and the model used for the correction are both calculated by MAD-X.
- Usually **both** are taken from the last "twiss" table and are therefore the same.
- The algorithms always minimize the r.m.s. and correct to the zero orbit.

In an online orbit correction procedure the orbit and the model obviously come from different sources and the correction to a (non-zero) target orbit is implemented in a rather straightforward way [5].

In this note I shall describe a modification in MAD-X which allows to use a model for the correction that is calculated independently from the calculated, perturbed orbit. A further change allows to correct to a pre-computed target orbit rather than to the zero orbit.

## 2 Orbit tables

To that purpose it is possible to define and store different types of tables for further use. As many tables as needed of each type can be computed and stored.

### 2.1 Table with perturbed orbit

As in the original implementation, the perturbed orbit is computed with a 'twiss' command, normally following a definition of imperfections (alignment errors, dipole errors etc.) generating the orbit. This table can be given a name with the 'table' specifier on the twiss command and it will then be available for further use. E.g.:

```
Twiss, table=myorbit, sequence=lhcb1;
```

### 2.2 Table with correction model

In the original implementation the (linear) model used by the correction algorithms was taken from the last computed twiss table, which also contained the perturbed orbit. Although in most cases this suffices the purpose of orbit corrections in MAD-X, it may be inadequate to simulate an orbit correction in a control system, in particular when the orbit is very large or when the effect of (unknown) deficiencies in the model are studied. Usually the model used in the control system is one of an unperturbed machine (e.g. LEP) and only when the model is well enough known, this will result in satisfactory results. In this implementation it is now possible to compute the model e.g.:

```
Twiss, table=mymodel, sequence=lhcb1;
```

with known or without any imperfections and store it for use under the name *mymodel*. For some studies it may be very useful to use different models to correct the same orbit and this is allowed by this modification.

### 2.3 Table with target orbit

Presently, in the MAD-X program the closed orbit is corrected to the zero orbit. However, it may be required to correct the orbit to follow a well defined orbit that may not be the reference orbit. In the operation of accelerators this is a well known requirement. In storage rings the desired orbit is not always the zero orbit, e.g. in case of beam separation using pretzl orbits or the LHC with crossing angle and separation bumps. Instead of treating the orbit correction and the formation of the bumps separately one could request the wanted orbit as the target for the orbit correction. In this implementation this becomes possible in MAD-X (from version 1.11 or later). It first requires the computation of the target

orbit e.g.:

```
Twiss, table=mytarget, sequence=lhcb1;
```

which has all the wanted features.

### 3 MAD-X directives

To allow the use of different models or target orbits, the CORRECT command in MAD-X has been extended. A standard orbit correction in the horizontal plane is requested by:

```
correct,mode=micado,ncorr=100,plane=x,error=10e-4;
```

using the MICADO algorithm and 100 correctors. Both, the model and the orbit are taken from the last *twiss* table created during a MAD-X run.

The new features can be invoked like:

```
correct,mode=micado,ncorr=100,plane=x,error=10e-4,  
orbit=myorbit,model=mymodel,target=mytarget;
```

If the *model* or *orbit* directives are omitted, the corresponding information is taken from the last *twiss* table as a default. If *target* is omitted, the orbit is corrected to the zero orbit.

### 4 Example

For illustration I have included a more complicated but realistic example in appendix A. This example can be found in the section with examples in [2]. It treats ring1 of the LHC for the collision optics.

In the preliminaries the input is set up and some classes of 'correctors' are removed from the sequence. These are part of the injection and dumping system and are declared as orbit correctors in the input since no special class is foreseen for such elements. Their presence may obscure orbit correction and they must be taken out or disabled before a LHC orbit correction is performed.

In the second step the model is provided by calculating the table *my\_model* using the *twiss* command for a unperturbed LHC without crossing angles or bumps.

In the next step two potential target orbits are created and saved which are the normal LHC machine with crossing bumps of different size (crossing angle) added (*my\_target1* and *my\_target2*). The horizontal and vertical orbits are shown in Figs. 1 - 2. The crossing angle bumps are precomputed and can be activated and scaled with the variables *on\_x1* etc.

The orbit to be corrected is generated by misalignment of quadrupoles of the class=MQ. In the horizontal plane the misalignments have a Gaussian distribution with a r.m.s. of 0.07 mm and in the vertical plane a r.m.s. of 0.10 mm. These orbits are stored as *my\_orbit* and shown in Fig. 3. In the first correction the orbit is corrected to the zero orbit and the results are shown in Fig.4. In the second orbit correction these orbits are corrected to follow *my\_target1*. Some magnets have to be disabled to avoid problems with the algorithms. They are dedicated magnets to compensate the experimental spectrometer magnets in IP2 and IP8. To check and plot the resulting orbits, the *twiss* command is used. The identical misalignment errors are specified as for *my\_orbit*<sup>1</sup> and the computed correctors are taken from the sequence of elements. The results of the corrections are shown in Fig. 5. The closed orbit from Fig. 3 is nicely corrected and the crossing angle bumps are established as requested.

The same procedure is repeated for the second set of target orbits *my\_target2* and the results are presented in Fig. 6. This time the orbit follows the bumps with different amplitudes as desired.

---

<sup>1</sup>The smart MAD user may realise that this is not required in this particular example but was added to improve the clarity.




Figure 1: Target orbit 1 in horizontal and vertical plane.

## 5 Specific features and possible pitfalls

As different twiss tables can now be used for the same correction process, some issues must be remembered to get correct and consistent results.

### 5.1 Change of sequence

The twiss tables in MAD-X have a specific feature which makes their use more subtle. They contain pointers into the parent sequence and these are used at run time, e.g. by the orbit correction routines. These pointers may be different (e.g. due to different sequences at creation time) for the different tables used for the correction, and this could lead to unpredictable results in some cases. The *USE* of a new or modified sequence between the creation of the different tables can create such problems and should therefore be avoided. *Using* a sequence to reset errors is therefore not recommended.

Although obvious, it should be said that tables corresponding to different sequences (e.g. LHC ring1 and ring2) must not be mixed on the same CORRECT command.




Figure 2: Target orbit 2 in horizontal and vertical plane.

## 5.2 Activation and de-activation of monitors and correctors

The activation and de-activation of monitors and correctors is controlled with the commands USEMONITOR and USEKICK. This information is stored in the sequence (see above) and not in any of the twiss tables. The element names are used for activation and de-activation. Once selected with USEMONITOR and USEKICK, the elements will therefore retain their selection flag unless the sequence is *used* (not recommended !) or reset using again the USEMONITOR and USEKICK commands.





Figure 3: Perturbed orbit for correction.




Figure 4: Orbit corrected to zero 1 in horizontal and vertical plane.




Figure 5: Orbit corrected to target 1 in horizontal and vertical plane.






Figure 6: Orbit corrected to target 2 in horizontal and vertical plane.

## References

- [1] W. Herr; *Implementation of new orbit correction procedures in the MAD-X program*, CERN-SL-2002-48 (AP) (2002).
- [2] *The MAD-X Home Page, version February 2003*,  
<http://frs.home.cern.ch/frs/Xdoc/mad-X.html>.
- [3] W. Herr, T. Limberg, J. Miles, R. Schmidt; *A new Closed Orbit Correction procedure for the CERN-SPS*;  
1989 Particle Accelerator Conference, Chicago, USA, March 20-23
- [4] D. Brandt, W. Herr, J. Miles and R. Schmidt; *A New Closed Orbit Correction Procedure for the CERN SPS and LEP*;  
Nucl. Instr. and Meth., **A293** (1990) 305. and CERN/SPS/89-48 (AMS).
- [5] W. Herr; *Algorithms and procedures used in the orbit correction packages COCU*;  
CERN/SL/95-07 (AP).

## 6 Appendix A

```
Option, warn,info,echo;
Assign, print=term;

System,"ln -fns /afs/cern.ch/eng/lhc/optics/V6.4 db4";

Title, "LHC 6.4 with errors and orbit correction using MADX";

// CALL IN SELECTION, SEQUENCE and OPTICS FILES;
Option, -echo,warn,-info;

call,file="db4/V6.4.seq";
call, file = "db4/V6.4.coll.str";

Beam, particle = proton, sequence=lhcb1, energy = 450.0,
      NPART=1.05E11, size=      4.5e-4 ;

// First make a clean machine as the model ...
Use, period=lhcb1;
on_x1 = 0.0;
on_x2 = 0.0;
on_x5 = 0.0;
on_x8 = 0.0;
on_alice = 0.0;
on_lhcb = 0.0;
select, flag=twiss, clear;
select, flag=twiss, column=name,s,x,px,y,py,betx,bety,alfx,alfy;
TWISS,table=my_model,sequence=lhcb1;

// Now a machine with the bumps as a target orbit
on_x1 = 1.0;
on_x2 = 1.0;
on_x5 = 1.0;
on_x8 = 1.0;
on_alice = 0.0;
on_lhcb = 0.0;
select, flag=twiss, clear;
select, flag=twiss, column=name,s,x,px,y,py,betx,bety,alfx,alfy;
TWISS,table=my_target1,sequence=lhcb1;
plot,table=my_target1,title="Target orbit 1 x...",
noline=true,vmin=-0.012,vmax=0.012,haxis=s,vaxis=x;
plot,table=my_target1,title="Target orbit 1 y...",
noline=true,vmin=-0.012,vmax=0.012,haxis=s,vaxis=y;

// Now a machine with the modified bumps as another target orbit
on_x1 = 1.0;
on_x2 = 0.5;
on_x5 = 1.0;
```

```

on_x8 = 0.3;
on_alice = 0.0;
on_lhcb = 0.0;
select, flag=twiss, clear;
select, flag=twiss, column=name,s,x,px,y,py,betx,bety,alfx,alfy;
TWISS,table=my_target2,sequence=lhcb1;
plot,table=my_target2,title="Target orbit 2 x...",
noline=true,vmin=-0.012,vmax=0.012,haxis=s,vaxis=x;
plot,table=my_target2,title="Target orbit 2 y...",
noline=true,vmin=-0.012,vmax=0.012,haxis=s,vaxis=y;

// Switch off again the bumps and mis-align the machine
on_x1 = 0.0;
on_x2 = 0.0;
on_x5 = 0.0;
on_x8 = 0.0;
on_alice = 0.0;
on_lhcb = 0.0;

!*****
eoption, add=false, seed=22021955;
Select, flag=ERROR, clear;
Select, flag=ERROR, class=MQ;
ealign, dy:=tgauss(2.5)*0.100e-3, dx:=tgauss(2.5)*0.070e-3;
!*****

select, flag=twiss, clear;
select, flag=twiss, column=name,s,x,px,y,py,betx,bety,alfx,alfy;
twiss, table=my_orbit,sequence=lhcb1;
plot,table=my_orbit,title="Uncorrected orbit x...",
noline=true,vmin=-0.012,vmax=0.012,haxis=s,vaxis=x;
plot,table=my_orbit,title="Uncorrected orbit y...",
noline=true,vmin=-0.012,vmax=0.012,haxis=s,vaxis=y;

// Prepare first correction to zero orbit
USEKICK,CLASS=MBLW, status=OFF; ! IR8 spectrometer magnets
USEKICK,CLASS=MBXWH, status=OFF; ! IR8 spectrometer magnets
USEKICK,CLASS=MBXWS, status=OFF; ! IR8 spectrometer magnets
USEKICK,CLASS=MBXWT, status=OFF;
USEKICK,CLASS=MSDB, status=OFF;
USEKICK,CLASS=MKA, status=OFF;
USEKICK,CLASS=MCBYH, status=OFF;
USEKICK,CLASS=MCBYV, status=OFF;
! USEKICK,CLASS=MCBX, status=OFF;
USEKICK,class=MKI, status=OFF;
USEKICK,class=MKA, status=OFF;
USEKICK,class=MKD, status=OFF;
USEKICK,class=MKQ, status=OFF;
USEKICK,class=MSDA, status=OFF;
USEKICK,class=MSDB, status=OFF;

```

```

USEKICK,class=MSDB2, status=OFF;
USEKICK,class=MSDC, status=OFF;
USEKICK,class=MSIA, status=OFF;
USEKICK,class=MSIB, status=OFF;
USEKICK,class=MBXWS, status=OFF;
USEKICK,RANGE=MBAW.R2, status=OFF; ! IP2 spectrometer compensation
USEKICK,RANGE=MBWMD.1L2, status=OFF; ! IP2 spectrometer compensation

coption,print=2;

CORRECT, FLAG=ring,orbit=my_orbit,model=my_model,
        MODE=micado,cond=1,NCORR=300,ERROR=1E-6,PLANE=y,
        mlist=my.tab,clist='cy.tab', RESOUT=1,corzero=1;
CORRECT, FLAG=ring,orbit=my_orbit,model=my_model,
        MODE=micado,cond=1,NCORR=300,ERROR=1E-6,PLANE=x,
        mlist=mx.tab,clist='cx.tab', RESOUT=1,corzero=1;
!*****

on_x1 = 0.0;
on_x2 = 0.0;
on_x5 = 0.0;
on_x8 = 0.0;
on_alice = 0.0;
on_lhcb = 0.0;

// mis-align the machine with the same errors for test of
// the correction

eoption, add=false, seed=22021955;
Select, flag=ERROR, clear;
Select, flag=ERROR, class=MQ;
ealign, dy:=tgauss(2.5)*0.100e-3, dx:=tgauss(2.5)*0.070e-3;

select, flag=twiss, clear;
select, flag=twiss, column=name,s,x,px,y,py,betx,bety,alfx,alfy;
TWISS,file=result.file,sequence=lhcb1;
plot,title="Corrected orbit x...",
noline=true,vmin=-0.012,vmax=0.012,haxis=s,vaxis=x;
plot,title="Corrected orbit y...",
noline=true,vmin=-0.012,vmax=0.012,haxis=s,vaxis=y;

// Prepare first correction to target orbit my_target1

CORRECT, FLAG=ring,orbit=my_orbit,target=my_target1,model=my_model,
        MODE=micado,cond=1,NCORR=300,ERROR=1E-6,PLANE=y,
        mlist=my.tab,clist='cy.tab', RESOUT=1,corzero=1;
CORRECT, FLAG=ring,orbit=my_orbit,target=my_target1,model=my_model,
        MODE=micado,cond=1,NCORR=300,ERROR=1E-6,PLANE=x,
        mlist=mx.tab,clist='cx.tab', RESOUT=1,corzero=1;
!*****

```

```

on_x1 = 0.0;
on_x2 = 0.0;
on_x5 = 0.0;
on_x8 = 0.0;
on_alice = 0.0;
on_lhcb = 0.0;

// mis-align the machine with the same errors for test of
// the correction

eoption, add=false, seed=22021955;
Select, flag=ERROR, clear;
Select, flag=ERROR, class=MQ;
ealign, dy:=tgauss(2.5)*0.100e-3, dx:=tgauss(2.5)*0.070e-3;

select, flag=twiss, clear;
select, flag=twiss, column=name,s,x,px,y,py,betx,bety,alfx,alfy;
TWISS,file=result.file,sequence=lhcb1;
plot,title="Corrected orbit target 1 x...",
noline=true,vmin=-0.012,vmax=0.012,haxis=s,vaxis=x;
plot,title="Corrected orbit target 1 y...",
noline=true,vmin=-0.012,vmax=0.012,haxis=s,vaxis=y;

// Prepare second correction to target orbit my_target2

USEKICK,CLASS=MCBYH, status=ON; ! Mark
USEKICK,CLASS=MCBYV, status=ON; ! Mark
USEKICK,CLASS=MCBX, status=ON; ! Mark

USEKICK,RANGE=mcbyh.a4l1.b1, status=OFF;
USEKICK,RANGE=mcbxa.3r2, status=OFF;
USEKICK,RANGE=mcbx.2l2, status=OFF;

CORRECT, FLAG=ring,orbit=my_orbit,target=my_target2,model=my_model,
MODE=micado,cond=1,NCORR=300,ERROR=1E-6,PLANE=x,
mlist=mx.tab,clist='cx.tab',RESOUT=1,corzero=1;
CORRECT, FLAG=ring,orbit=my_orbit,target=my_target2,model=my_model,
MODE=micado,cond=1,NCORR=300,ERROR=1E-6,PLANE=y,
mlist=my.tab,clist='cy.tab', RESOUT=1,corzero=1;

!*****
on_x1 = 0.0;
on_x2 = 0.0;
on_x5 = 0.0;
on_x8 = 0.0;
on_alice = 0.0;
on_lhcb = 0.0;

// mis-align the machine with the same errors for test of

```

```

// the correction

eoption, add=false, seed=22021955;
Select, flag=ERROR, clear;
Select, flag=ERROR, class=MQ;
ealign, dy:=tgauss(2.5)*0.100e-3, dx:=tgauss(2.5)*0.070e-3;
!*****

select, flag=twiss, clear;
select, flag=twiss, column=name,s,x,px,y,py,betx,bety,alfx,alfy;
TWISS,file=result.file,sequence=lhcb1;
plot,title="Corrected orbit target 2 x...",
noline=true,vmin=-0.012,vmax=0.012,haxis=s,vaxis=x;
plot,title="Corrected orbit target 2 y...",
noline=true,vmin=-0.012,vmax=0.012,haxis=s,vaxis=y;

stop;

```